



同順位のあるウィルコクソン順位和統計量とアンサ  
リ・ブラッドレイ統計量の分布計算プログラム

メタデータ	言語: jpn 出版者: 公開日: 2013-11-22 キーワード (Ja): キーワード (En): 作成者: 清水, 行雄 メールアドレス: 所属:
URL	<a href="https://doi.org/10.24729/00007861">https://doi.org/10.24729/00007861</a>

# 同順位のあるウィルコクソン順位和統計量とアンサリ・ブラッドレイ統計量の分布計算プログラム

清水行雄\*

## The Calculation Programs of the Distribution of the Wilcoxon Rank Sum Statistic and the Ansari-Bradley Statistic when Ties are Present

Yukio SHIMIZU\*


### ABSTRACT

The BASIC programs are given for calculating the small-sample distribution of the Wilcoxon rank sum statistic and the Ansari-Bradley statistic when ties are present.

Key Words: Nonparametric test, Wilcoxon rank sum test, Ansari-Bradley test, Ties, Critical values.

### 1. 検定と統計量

$F(x)$  より確率変数  $X_1, \dots, X_m, G(y)$  より確率変数  $Y_1, \dots, Y_n, N = m+n$  個を観測したとする。さらに、合併標本での同順位のパターンが



であるとする。

$H_0: F(x) = G(x)$  対  $H_1: F(x) > G(x)$  ( $F(x) < G(x), F(x) \neq G(x)$ ) を検定する。そのための統計量として、同順位のあるとき、ウィルコクソン順位和統計量

$$U^* = \sum_{i=1}^m R_i^* - \frac{m(m+1)}{2}$$

がある。ここで  $R_i^*$  は合併標本での  $X_i$  の中間順位である。

次に、 $G(y) = F(\theta y)$  であるとして、 $H_0: \theta = 1$  対  $H_1: \theta > 1$  ( $\theta < 1, \theta \neq 1$ ) を検定したいとする。同順位のあるとき、アンサリ・ブラッドレイ統計量は

$$W^* = \sum_{i=1}^m \left( \frac{N+1}{2} - \left| R_i^* - \frac{N+1}{2} \right| \right)$$

であろう。やはり  $R_i^*$  は合併標本での  $X_i$  の中間順位である。

$U^*$  の分布も、 $W^*$  の分布もともに漸近的に正規分布

をする。同順位がない場合は Wilcoxon, Katti and Wilcoxon(1973), Ansari and Bradley (1960)などに有意確率を求めるための数表があるが、同順位のある場合の数表はみあたらない。また、ウィルコクソン順位和検定とアンサリ・ブラッドレイ検定のためのプログラムは杉山・牛沢(1984)、白旗(1987)にあるが、確率計算には正規近似がもちいられている。本報告では、小標本のときの  $U^*$  と  $W^*$  の帰無分布の確率を精度よく計算するプログラムを作成した。

### 2. BASIC 言語によるプログラム

(a) プログラム tiedranksum

標本の大きさ  $m, n$  とウィルコクソン順位和統計量の実現値  $u$  または  $u'$  を入力したとき  $P_r\{U^* \leq u | H_0\} = \alpha, P_r\{U^* \geq u' | H_0\} = \alpha$  を出力し、また  $\alpha$  を入力したとき  $P_r\{U^* \leq u | H_0\} = \alpha, P_r\{U^* \geq u' | H_0\} = \alpha$  なる  $u, u'$  を出力する。

(b) プログラム TRSSTATISTIC

観測データよりウィルコクソン順位和統計量の実現値を計算するプログラムで、同順位のパターンも出力する。

(c) プログラム tiedansaribradley

標本の大きさ  $m, n$  とアンサリ・ブラッドレイ統計量の実現値  $w$  または  $w'$  を入力したとき  $P_r\{W^* \leq w | H_0\} = \alpha, P_r\{W^* \geq w' | H_0\} = \alpha$  を出力し、また  $\alpha$  を入力したとき  $P_r\{W^* \leq w | H_0\} = \alpha, P_r\{W^* \geq w' | H_0\} = \alpha$  なる  $w, w'$  を出力する。

(d) プログラム TABSTATISTIC

観測データよりアンサリ・ブラッドレイ統計量の実現値を計算するプログラムで、同順位のパターンも出力する。

1991年4月10日受理

\* 一般教養科 (Department of Liberal Arts)

## プログラム tiedranksum

```

100 'WILCOXON RANK SUM TEST WITH TIES
110 '*** file name tiedranksum ***
120 DEFDBL A,C
130 READ N1,N2
140 N=N1+N2:MU=N1*N2*.5
150 IF N1<N2 THEN NMIN=N1 ELSE NMIN=N2
160 C=1#
170 FOR I=1 TO NMIN
180 C=C*(N-I+1#)/I
190 NEXT I
200 READ E
210 DIM D(E),DD(E),A(N),B(N1),C(4*MU)
220 FOR I=1 TO E
230 READ D(I)
240 DD(I)=DD(I-1)+D(I)
250 NEXT I
260 IF DD(E)<>N THEN PRINT "DATA ?":GOTO 1200
270 LPRINT "** m=";N1;" n=";N2;" ** d=";
280 FOR I=1 TO E
290 LPRINT D(I);
300 NEXT I
310 LPRINT:LPRINT "          ** r=";
320 FOR I=1 TO E
330 A=DD(I-1)+.5*(D(I)+1#)
340 FOR J=DD(I-1)+1 TO DD(I)
350 A(J)=A
360 NEXT J
370 LPRINT A(DD(I));
380 NEXT I
390 LPRINT:LPRINT "          **";
400 INPUT "input U or Alpha ( U/A )";Ws
410 IF Ws<>"U" THEN 470
420 INPUT "U=";V
430 IF V<0 OR V>2*MU THEN 420
440 IF INT(2*V)<>2*V THEN 420
450 LPRINT " U= ";V
460 GOTO 510
470 IF Ws<>"A" THEN 400
480 INPUT "alpha=";X
490 IF X<0 OR X>.5 THEN 480
500 LPRINT " alpha= ";X
510 FOR I=1 TO N1
520 B(I)=I
530 NEXT I
540 U=-N1*(N1+1)
550 FOR I=1 TO N1
560 U=U+A(B(I))*2
570 NEXT I
580 C(U)=C(U)+1#
590 FOR I=1 TO N1
600 IF B(N1-I+1)=N-I+1 THEN 670
610 B(N1-I+1)=B(N1-I+1)+1
620 IF I=1 THEN 540
630 FOR J=1 TO I-1
640 B(N1-I+J+1)=B(N1-I+J)+1
650 NEXT J
660 GOTO 540
670 NEXT I
680 IF N1=N2 THEN 760
690 E2=INT(E*.5)
700 FOR I=1 TO E2
710 IF D(I)<>D(E-I+1) THEN 740
720 NEXT I
730 GOTO 760
740 Ss="AS"
750 IF Ws="U" AND V>MU THEN 1040
760 ALPHA=0#:Q=0:U21=0
770 FOR U=0 TO 2*MU
780 U2=U*.5
790 IF C(U)=0# THEN 1030
800 ALPHA=ALPHA+C(U)/C
810 IF Ss="AS" THEN 920
820 IF Ws="A" THEN 860
830 IF V>MU THEN V=2*MU-V
840 IF U2=V THEN 890
850 IF U2>V THEN 880 ELSE 1010

```

```

860 IF ALPHA<=X THEN 1010
870 IF U=0 THEN U21=-.5
880 LPRINT "Pr { U≤";U21;" } = Pr { U≥";2*MU-U21;" } =";Q
890 P=ALPHA
900 LPRINT "Pr { U≤";U2;" } = Pr { U≥";2*MU-U2;" } =";P
910 GOTO 1200
920 IF W$="A" THEN 950
930 IF U2=V THEN 980
940 IF U2>V THEN 970 ELSE 1010
950 IF ALPHA<=X THEN 1010
960 IF U=0 THEN U21=-.5
970 LPRINT "Pr { U≤";U21;" } =";Q
980 P=ALPHA
990 LPRINT "Pr { U≤";U2;" } =";P
1000 IF W$="U" THEN 1200 ELSE 1040
1010 Q=ALPHA
1020 U21=U2
1030 NEXT U
1040 ALPHA=0#:Q=0:U21=2*MU
1050 FOR U=4*MU TO 2*MU STEP -1
1060 U2=U*.5
1070 IF C(U)=0# THEN 1190
1080 ALPHA=ALPHA+C(U)/C
1090 IF W$="A" THEN 1110
1100 IF U2<=V THEN 1130 ELSE 1180
1110 IF ALPHA<=X THEN 1180
1120 IF U=4*MU THEN U21=2*MU+.5
1130 P=ALPHA
1140 LPRINT "          Pr { U≥";U2;" } =";P
1150 IF W$="U" AND U2=V THEN 1200
1160 LPRINT "          Pr { U≥";U21;" } =";Q
1170 GOTO 1200
1180 Q=ALPHA:U21=U2
1190 NEXT U
1200 END
2000 DATA 9,7
2010 DATA 11
2020 DATA 1,2,1,2,2,1,1,3,1,1,1

```

## 出力例

```

** m= 6  n= 6  ** d= 2  1  3  2  1  2  1
** r= 1.5  3  5  7.5  9  10.5  12
** alpha= .01
Pr { U≤ 2.5 } = Pr { U≥ 33.5 } = 7.57576E-03
Pr { U≤ 4 } = Pr { U≥ 32 } = .0108225
** m= 6  n= 6  ** d= 2  1  3  2  1  2  1
** r= 1.5  3  5  7.5  9  10.5  12
** alpha= .05
Pr { U≤ 7.5 } = Pr { U≥ 28.5 } = .0454545
Pr { U≤ 8 } = Pr { U≥ 28 } = .0584416
** m= 9  n= 7  ** d= 1  2  1  2  2  1  1  3  1  1  1
** r= 1  2.5  4  5.5  7.5  9  10  12  14  15  16
** alpha= .01
Pr { U≤ 10 } = 9.79021E-03
Pr { U≤ 10.5 } = .0117133
Pr { U≥ 53 } = .0105769
Pr { U≥ 53.5 } = 8.21678E-03
** m= 9  n= 7  ** d= 1  2  1  2  2  1  1  3  1  1  1
** r= 1  2.5  4  5.5  7.5  9  10  12  14  15  16
** alpha= .05
Pr { U≤ 15.5 } = .0472028
Pr { U≤ 16 } = .0539336
Pr { U≥ 47 } = .0521853
Pr { U≥ 47.5 } = .0465035
** m= 9  n= 7  ** d= 1  2  1  2  2  1  1  3  1  1  1
** r= 1  2.5  4  5.5  7.5  9  10  12  14  15  16
** U= 43
Pr { U≥ 43 } = .120717

```

## プログラム TRSSTATISTIC

```

100 'WILCOXON RANK SUM TEST WITH TIES
110 '*** file name TRSSTATISTIC ***
120 READ N1,N2
130 LPRINT "** m= ";N1;" n= ";N2;"**"
140 N=N1+N2
150 DIM X(N),D(N)
160 LPRINT "X = ";
170 FOR I=1 TO N1
180 READ X(I)
190 LPRINT X(I);
200 D(I)=1
210 NEXT I
220 LPRINT:LPRINT "Y = ";
230 FOR I=N1+1 TO N
240 READ X(I)
250 LPRINT X(I);
260 D(I)=1
270 NEXT I
280 LPRINT
290 V=0
300 FOR I=1 TO N1
310 FOR J=N1+1 TO N
320 IF X(I)>X(J) THEN V=V+1
330 IF X(I)=X(J) THEN V=V+.5
340 NEXT J
350 NEXT I
360 FOR I=1 TO N-1
370 FOR J=I+1 TO N
380 IF X(I)>X(J) THEN SWAP X(I),X(J)
390 NEXT J
400 NEXT I
410 E=1
420 FOR I=1 TO N-1
430 IF X(I)=X(I+1) THEN D(E)=D(E)+1:GOTO 450
440 E=E+1
450 NEXT I
460 LPRINT "e= ";E
470 LPRINT "d= ";
480 FOR I=1 TO E
490 LPRINT D(I);
500 NEXT I
510 LPRINT
520 IF N1=N2 THEN 570
530 E2=INT(E*.5)
540 FOR I=1 TO E2
550 IF D(I)<>D(E-1+1) THEN 590
560 NEXT I
570 LPRINT "Ux= ";V;" Uy= ";N1*N2-V
580 GOTO 600
590 LPRINT "Ux= ";V
600 END
1000 DATA 9,7
1010 DATA 2.4,2.0,2.1,3.1,2.2,2.4,2.8,1.9,1.3
1020 DATA 1.9,2.4,2.0,1.5,1.5,2.6,1.7

```

## 出 力 例

```

** m= 9 n= 7 **
X = 2.4 2 2.1 3.1 2.2 2.4 2.8 1.9 1.3
Y = 1.9 2.4 2 1.5 1.5 2.6 1.7
e= 11
d= 1 2 1 2 2 1 1 3 1 1 1
Ux= 43

```

## プログラム tiedansaribradley

```

100 'ANSARI-BRADLEY TEST WITH TIES
110 '*** file name      tiedansaribradley ***
120 DEFDBL A,C
130 READ N1,N2
140 N=N1+N2
150 IF N1<N2 THEN NMIN=N1 ELSE NMIN=N2
160 C=1#
170 FOR I=1 TO NMIN
180 C=C*(N-I+1#)/1
190 NEXT I
200 READ E
210 DIM D(E),DD(E),R(N),A(N),B(N1)
220 FOR I=1 TO E
230 READ D(I)
240 DD(I)=DD(I-1)+D(I)
250 NEXT I
260 IF DD(E)<>N THEN PRINT "DATA ?":GOTO 1290
270 LPRINT "** m=";N1;" n=";N2;" ** d=";
280 FOR I=1 TO E
290 LPRINT D(I);
300 NEXT I
310 LPRINT:LPRINT "          ** r=";
320 FOR I=1 TO E
330 R=DD(I-1)+.5*(D(I)+1)
340 FOR J=DD(I-1)+1 TO DD(I)
350 R(J)=R
360 NEXT J
370 LPRINT R(DD(I));
380 NEXT I
390 FOR I=1 TO N
400 A(I)=(N+1#)/2#-ABS(R(I)-(N+1#)/2#)
410 A=A+A(I)
420 NEXT I
430 MU=A/N*N1
440 FOR I=1 TO N-1
450 FOR J=I+1 TO N
460 IF A(I)>A(J) THEN SWAP A(I),A(J)
470 NEXT J,I
480 FOR I=1 TO N1
490 MIN=MIN+A(I)
500 MAX=MAX+A(N-I+1)
510 NEXT I
520 DIM C(2*MAX)
530 LPRINT:LPRINT "          **";
540 INPUT "input W or Alpha ( W/A )";W#
550 IF W#<>"W" THEN 610
560 INPUT "W=";V
570 IF V<MIN OR V>MAX THEN 560
580 IF INT(2*V)<>2*V THEN 560
590 LPRINT " W= ";V
600 GOTO 650
610 IF W#<>"A" THEN 540
620 INPUT "alpha=";X
630 IF X<0 OR X>.5 THEN 620
640 LPRINT " alpha= ";X
650 FOR I=1 TO N1
660 B(I)=I
670 NEXT I
680 U=0
690 FOR I=1 TO N1
700 U=U+A(B(I))*2
710 NEXT I
720 C(U)=C(U)+1#
730 FOR I=1 TO N1
740 IF B(N1-I+1)=N-I+1 THEN 810
750 B(N1-I+1)=B(N1-I+1)+1
760 IF I=1 THEN 680
770 FOR J=1 TO I-1
780 B(N1-I+J+1)=B(N1-I+J)+1
790 NEXT J
800 GOTO 680
810 NEXT I
820 IF N1=N2 THEN 850
830 S#="AS"
840 IF W#="W" AND V>MU THEN 1130
850 ALPHA=0#:Q=0:U21=MIN

```

```

860 FOR U=2*MIN TO 2*MU
870 U2=U*.5
880 IF C(U)=0# THEN 1120
890 ALPHA=ALPHA+C(U)/C
900 IF S$="AS" THEN 1010
910 IF W$="A" THEN 950
920 IF V>MU THEN V=2*MU-V
930 IF U2=V THEN 980
940 IF U2>V THEN 970 ELSE 1100
950 IF ALPHA<=X THEN 1100
960 IF U=2*MIN THEN U21=MIN-.5
970 LPRINT "Pr { W≤";U21;" } = Pr { W≥";2*MU-U21;" } =";Q
980 P=ALPHA
990 LPRINT "Pr { W≤";U2;" } = Pr { W≥";2*MU-U2;" } =";P
1000 GOTO 1290
1010 IF W$="A" THEN 1040
1020 IF U2=V THEN 1070
1030 IF U2>V THEN 1060 ELSE 1100
1040 IF ALPHA<=X THEN 1100
1050 IF U=2*MIN THEN U21=MIN-.5
1060 LPRINT "Pr { W≤";U21;" } =";Q
1070 P=ALPHA
1080 LPRINT "Pr { W≤";U2;" } =";P
1090 IF W$="W" THEN 1290 ELSE 1130
1100 Q=ALPHA
1110 U21=U2
1120 NEXT U
1130 ALPHA=0#:Q=0:U21=MAX
1140 FOR U=2*MAX TO 2*MU STEP -1
1150 U2=U*.5
1160 IF C(U)=0# THEN 1280
1170 ALPHA=ALPHA+C(U)/C
1180 IF W$="A" THEN 1200
1190 IF U2<=V THEN 1220 ELSE 1270
1200 IF ALPHA<=X THEN 1270
1210 IF U=2*MAX THEN U21=MAX+.5
1220 P=ALPHA
1230 LPRINT "          Pr { W≥";U2;" } =";P
1240 IF W$="W" AND U2=V THEN 1290
1250 LPRINT "          Pr { W≥";U21;" } =";Q
1260 GOTO 1290
1270 Q=ALPHA:U21=U2
1280 NEXT U
1290 END
2000 DATA 9,7
2010 DATA 11
2020 DATA 1,2,1,2,2,1,1,3,1,1,1

```

## 出力例

```

** m= 6  n= 6  ** d= 2 1 3 2 1 2 1
** r= 1.5 3 5 7.5 9 10.5 12
** alpha= .01
Pr { W≤ 14 } = Pr { W≥ 28 } = 7.57576E-03
Pr { W≤ 14.5 } = Pr { W≥ 27.5 } = .0183983
** m= 6  n= 6  ** d= 2 1 3 2 1 2 1
** r= 1.5 3 5 7.5 9 10.5 12
** alpha= .05
Pr { W≤ 15.5 } = Pr { W≥ 26.5 } = .0367965
Pr { W≤ 16 } = Pr { W≥ 26 } = .0519481
** m= 9  n= 7  ** d= 1 2 1 2 2 1 1 3 1 1 1
** r= 1 2.5 4 5.5 7.5 9 10 12 14 15 16
** alpha= .01
Pr { W≤ 29.5 } = 8.21678E-03
Pr { W≤ 30 } = .0109266
          Pr { W≥ 51 } = .0108392
          Pr { W≥ 51.5 } = 7.86713E-03
** m= 9  n= 7  ** d= 1 2 1 2 2 1 1 3 1 1 1
** r= 1 2.5 4 5.5 7.5 9 10 12 14 15 16
** alpha= .05
Pr { W≤ 32.5 } = .0464161
Pr { W≤ 33 } = .058479
          Pr { W≥ 48 } = .0582168
          Pr { W≥ 48.5 } = .0457168
** m= 9  n= 7  ** d= 1 2 1 2 2 1 1 3 1 1 1
** r= 1 2.5 4 5.5 7.5 9 10 12 14 15 16
** W= 42
          Pr { W≥ 42 } = .394493

```

## プログラム TABSTATISTIC

```

100 'ANSARI-BRADLEY TEST WITH TIES
110 '*** file name      TABSTATISTIC ***
120 READ N1,N2
130 LPRINT "m= ";N1;" n= ";N2;" **"
140 N=N1+N2
150 DIM X(N),Z(N),D(N),DD(N),R(N),A(N)
160 LPRINT "X = ";
170 FOR I=1 TO N1
180 READ X(I):LPRINT X(I);
190 Z(I)=1
200 NEXT I
210 LPRINT :LPRINT "Y = ";
220 FOR I=N1+1 TO N
230 READ X(I):LPRINT X(I);
240 NEXT I
250 LPRINT
260 FOR I=1 TO N-1
270 FOR J=I+1 TO N
280 IF X(I)>X(J) THEN SWAP X(I),X(J):SWAP Z(I),Z(J)
290 NEXT J, I
300 FOR I=1 TO N
310 D(I)=1
320 NEXT I
330 E=1
340 FOR I=1 TO N-1
350 IF X(I)=X(I+1) THEN D(E)=D(E)+1:GOTO 370
360 E=E+1
370 NEXT I
380 LPRINT "e= ";E
390 LPRINT "d= ";
400 FOR I=1 TO E
410 LPRINT D(I);
420 DD(I)=DD(I-1)+D(I)
430 NEXT I
440 LPRINT
450 FOR I=1 TO E
460 R=DD(I-1)+.5*(D(I)+1)
470 FOR J=DD(I-1)+1 TO DD(I)
480 R(J)=R
490 NEXT J, I
500 FOR I=1 TO N
510 A(I)=(N+1)/2-ABS(R(I)-(N+1)/2)
520 NEXT I
530 FOR I=1 TO N
540 A=A+A(I)*Z(I)
550 A1=A1+A(I)*(1-Z(I))
560 NEXT I
570 IF N1=N2 THEN 580 ELSE 600
580 LPRINT "Wx= ";A;"      Wy= ";A1
590 GOTO 610
600 LPRINT "Wx= ";A
610 END
1000 DATA 9,7
1010 DATA 2.4,2.0,2.1,3.1,2.2,2.4,2.8,1.9,1.3
1020 DATA 1.9,2.4,2.0,1.5,1.5,2.6,1.7

```

## 出力例

```

** m= 9 n= 7 **
X = 2.4 2 2.1 3.1 2.2 2.4 2.8 1.9 1.3
Y = 1.9 2.4 2 1.5 1.5 2.6 1.7
e= 11
d= 1 2 1 2 2 1 1 3 1 1 1
Wx= 42

```



PC 9801 RS による出力例の計算時間は,  $m = 9$ ,  $n = 7$  のとき(a)の場合 3分00秒, (c)の場合 2分55秒であった.

プログラムの作成にあたり, 宮本良雄先生に御助言, 御指導をいただきました. ここに記して深く感謝の意を表します.

### 参 考 文 献

- [1] Ansari, A.R. and Bradley, R.A. (1960), Rank-sum tests for dispersions, *A.M.S.*, 31, 1174-1189.
- [2] Hájek, J. (1969), *A Course in Nonparametric Statistics*, Holden-Day.
- [3] Klotz, J.H. (1966), The Wilcoxon, ties, and the computer, *J. A. S. A.*, 61, 772-787.
- [4] Lehmann, E.L. (1975), *Nonparametrics, Statistical Methods Based on Ranks*, Holden-Day.
- [5] 清水行雄, 宮本良雄 (1985), ノンパラメトリック統計量の分布計算プログラム, 大阪府立工業高等専門学校研究紀要, 19, 173-184.
- [6] 白旗慎吾 (1987), パソコン統計解析ハンドブック, IV, ノンパラメトリック編, 共立出版.
- [7] 杉山高一, 牛沢賢二 (1984), パソコンによる統計解析, 朝倉書店.
- [8] Wilcoxon, F., Katti, S.K. and Wilcox, R.A. (1973), Critical values and probability levels for the Wilcoxon rank sum test and the Wilcoxon signed rank test, *Selected Tables in Mathematical Statistics*, I, 171-259.
- [9] 山内二郎 (1972), 統計数値表, 日本規格協会.
- [10] 柳川 堯 (1982), ノンパラメトリック法, 培風館.